OUTPUT_CLAMP_REQUIREMENTS

```
* dsauersanjose@aol.com  12/17/08
* www.idea2ic.com
*
*                            US Patent # 5546045
```




```
.OPTIONS  method=trap

VIN       INP     0      SIN(6   6.1     1K )  AC 1m
VREF      VREF    0      DC  6
VCC       VCC     0      12
```

```
IP         VPNP   0      40u

B_RRIN     VRRIN  0      I = ( V(INP) - V(OUT) )/5000
RBYP       VRRIN  VIIN   1k

QP1        VP1C   VIIN   VPNPE   PNPV   1
QP2        VP2C   VREF   VPNPE   PNPV   1
QP3        VPNPE  VPNP   VCC     PNPV   1
QP4        VPNP   VPNP   VCC     PNPV   1

QN1        VCC    VP1C   VN2B    NPNV   1
QN2        VN2C   VN2B   0       NPNV   1
QN3        VST    VP2C   VN4B    NPNV   1
QN4        OUT    VN4B   0       NPNV   10
R1        VN2B    0      20K
R2        VP1C    VN2C   5K
R3        VP2C    VN2C   5K
R4        VN4B    0      10K

RST       VCC     VST     1
R8        OUT     VN5CC   300k
QN5       VN5CC   VN4B    VP2C    NPNV   1

CCOMP      OUT    VIIN    10p
BOTA       OUT    0       I = -200u

CBYP       VRRIN   VP2C   4p

.tran      1u     2m       0      1u
.model     NPNV   npn      BF=150 BR=5
.model     PNPV   pnp      BF=60  BR=5

.control
run
let Vshootthru =( V(vcc)-V(vst))*1e3
plot   V(Vshootthru) V(out)  title WithShootThur

alter R8 resistance = 300
run
let Vshootthru =( V(vcc)-V(vst))*1e3
plot   V(Vshootthru) V(out) title WithOutShootThur


.endc
.end
```

* =====================END===================================

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm

There are often hidden features inside an Op Amp design such as having an input that can exceed the supplies without phase reversal and having graceful death at low supply voltage. Another hidden feature has to do with addressing a supply current shoot thru problem.



When an output transistor like QN4 saturates, the transistor

driving the output transistor will try to dump a lot of current into the base of the output transistor. This has two negative effects. First, there will be a lot of charge in the base which will take some effort to remove in order for the output transistor to come out of sat. Second, the collector current of transistors like QN3 which are driving the output transistor will greatly increase.



This means every time the output hits one of the rails, an supply current spike will happen which is referred to a a shoot thru current.



This problem can be handle by building an output transistor saturation current limit. Transistor QN5 is a reverse connected transistor which is using the collector junction as an emitter and the emitter is being used as a collector. The performance of such a transistor is not great. But since QN6 and QN4 are both npns, the "emitter base" of QN5 will turn on exactly when the collector base of QN4 turns on. Some adjustments like R8 are sometimes needed to optimize this type of current limit.



Both the supply current shoot thru problem and the recovery from output saturation problem are improved. Features like this are not listed in the spec itself. But customers expect them anyway.